

Electronic supplementary material for manuscript

Experimental evidence that livestock grazing intensity affects cyclic vole population regulation processes.

Nacho Villar, Thomas Cornulier, Darren Evans, Robin Pakeman, Steve Redpath, Xavier Lambin

Statistical analyses

We used independent uninformative priors for all the parameters of the model, i.e., Normal ($\mu = 0, \sigma^2 = 10,000$) for intercepts and Uniform (0, 5) for the standard deviation of the random effects $\gamma_{1,i,j,t}, \gamma_{2,i}, \eta_{1,i,j,t}$ and $\eta_{2,i}$. For density dependence coefficients we used independent Normal ($\mu = 0, \sigma^2 = 1$) priors which means that the auto-regressive model was not *a priori* constrained to be stable (Congdon 2001). The initial conditions, i.e., the Spring and Fall abundance for the years $t = 0, -1$ were treated as latent (unobserved) data (Congdon 2001) with a Normal ($\mu = 0, \sigma^2 = 4$) prior. The model was fitted with OpenBUGS 2.2.0 using R's BRugs library (Lunn et al. 2009). We used 2,000,000 updates, of which one in 50 was kept to reduce memory storage needs, after a burn-in set of 50,000 updates. Computation time was under 3 hours on a 2.4 GHz CPU.

Figure S1 and Figure S2 show annual and full seasonal effects estimated per treatment, respectively. We found a high correlation between the autoregressive coefficients of the seasonal autoregressive model at successive time lags (Fig. S3), making inference about precise seasonal effects difficult. However, annual-level inference can be made by recombining the seasonal coefficients into the annual direct ($\Omega_{1,j}$) and delayed ($\Omega_{2,j}$) density-dependence parameters as

shown in Fig. 2 in the paper. The difference in annual parameters Ω_1 and Ω_2 between treatments remains obvious in spite of the correlations found in treatments 2 and 3 (Fig. S1).

References

- Congdon P (2001) Bayesian statistical modelling. Wiley Series in Probability and Statistics. John Wiley & Sons, Ltd., Chichester
- Lunn D, Spiegelhalter D, Thomas A, Best N (2009) The BUGS project: Evolution, critique and future directions. *Stat Med* 28:3049-3067
- Royama T (1992) Analytical population dynamics, vol 10. Population and Community Biology series. Chapman & Hall, New York

Fig. S1: effect of grazing intensity on direct (Ω_1) and delayed (Ω_2) density dependent regulation of field vole population growth, with 95 % posterior credible intervals. $T1 = 2.7 \text{ ewes ha}^{-1}$, $T2 = 0.9 \text{ ewe ha}^{-1}$, $T3 = \text{ungrazed}$. The figure also includes results for a fourth grazing treatment imposed in the experiment, which accounted for similar grazing intensity as T2 but with mixed cattle and sheep grazing (T4mix). These results for T4mix (open symbol and dotted lines) yield density-dependence estimates almost identical to T2, and are not reported in the manuscript for simplicity.




Fig. S2: estimated density-dependence (mean and 95 % posterior credible intervals of autoregressive coefficients) on the population growth rates in winter (upper row) and summer (lower row) for treatments T1, T2, T3, T4mix= mixed grazing. Direct density-dependence (no time lag) is shown in the first column. Estimated delayed density-dependence with a lag of 6, 12 and 18 months is shown in columns 2, 3 and 4 respectively.




Fig. S3: correlation matrices between the autoregressive parameters of the model. The values in the cells indicates the correlation coefficient between the Markov chains for each pair of parameters, showing large negative correlations between the autoregressive coefficients for successive time lags. Colours indicate the value of the correlation coefficients, from the minimum value shown in each panel (white) to the maximum (red = 1).




Fig. S4: correlation between the annual parameters Ω_1 and Ω_2 (direct and delayed density dependence respectively) derived from the seasonal parameters for treatment j as $\Omega_{1,j} = a_{1,j} + b_{1,j} + a_{2,j} + b_{2,j} + a_{1,j}b_{1,j}$ and $\Omega_{2,j} = a_{3,j} + b_{3,j} + a_{4,j} + b_{4,j} + a_{1,j}b_{3,j} + a_{3,j}b_{1,j} + a_{2,j}b_{2,j}$. Dots in red, green and blue colour are observations for treatments 1, 2 and 3, respectively. Black dots centered at every treatment show the posterior mean of the parameters and the dashed black lines delimit the 95 % posterior density estimate for each parameter. The parameter space described by Royama (1992) indicates unstable dynamics outside the triangle, stable dynamics with damped oscillations inside the triangle and above the parabola, and cyclic dynamics inside the parabola (cycle period length in years is given by grey numerals). Data for T4mix (mixed grazing) not shown for clarity (pattern is similar to treatment 2).




Table S1: Results of pair-wise comparisons between seasonal intercepts (Wi = winter; Su = summer) and direct (Ω_1) and delayed (Ω_2) density dependent parameters for all treatment combinations. The table includes means and standard deviations, Markov-Chain errors, values at 2.5 and 97.5 percentiles and median values.

Pairwise difference	mean	SD	MC_error	val2.5pc	median	val97.5pc
Intercept Wi T1-T2	0.371	0.916	0.011	-1.446	0.377	2.136
Intercept Wi T1-T3	-0.822	0.814	0.007	-2.515	-0.786	0.684
Intercept Wi T2-T3	-1.192	0.801	0.009	-2.853	-1.159	0.277
Intercept Su T1-T2	-1.473	2.285	0.025	-5.860	-1.521	3.208
Intercept Su T1-T3	0.796	1.744	0.016	-2.473	0.738	4.383
Intercept Su T2-T3	2.269	1.703	0.019	-1.074	2.268	5.677
Ω_1 T1-T2	0.233	0.394	0.004	-0.545	0.231	0.998
Ω_1 T1-T3	0.342	0.396	0.003	-0.440	0.344	1.111
Ω_1 T2-T3	0.109	0.375	0.003	-0.641	0.111	0.835
Ω_2 T1-T2	0.329	0.279	0.002	-0.203	0.324	0.897
Ω_2 T1-T3	0.676	0.303	0.002	0.094	0.673	1.282
Ω_2 T2-T3	0.347	0.295	0.002	-0.235	0.349	0.913

BUGS code for the analysis

```

model{
  # observation process
  for(Block in 1:6){
    for(Treat in 1:4){
      for(t in (k+1):T){
        # SPRING data
        NS[Block, t, Treat] ~ dbin(pS[Block, t, Treat], NQS[Block,
t, Treat]) # binomial with NQ attempts (25 VSI plots)
        cloglog(pS[Block, t, Treat])<- muS[Block, t, Treat]
        muS[Block, t, Treat] ~ dnorm(muS.proc[Block, t, Treat],
tauWi[Block, Treat])
        # FALL data
        NF[Block, t, Treat] ~ dbin(pF[Block, t, Treat], NQF[Block,
t, Treat])
        cloglog(pF[Block, t, Treat])<- muF[Block, t, Treat]
        muF[Block, t, Treat] ~ dnorm(muF.proc[Block, t, Treat],
tauSu[Block, Treat])
      }
      for(t in 1:k){
        muS[Block, t, Treat] ~ dnorm(0, 0.25)
        muF[Block, t, Treat] ~ dnorm(0, 0.25)
      }
    }
  }

  # system process
  for(Block in 1:6){
    for(t in (k+1):T){
      for(Treat in 1:4){
        # Equation for WINTER GROWTH:
        muS.proc[Block,t, Treat] <- InterceptWi[Treat] + a[Treat, 1]
        * muF[Block, t-1, Treat] + a[Treat, 2] * muS[Block, t-1, Treat] +
        a[Treat, 3] * muF[Block, t-2, Treat] + a[Treat, 4] * muS[Block, t-
2, Treat] + muF[Block, t-1, Treat] + BlockWi[Block]
        # Equation for SUMMER GROWTH:
        muF.proc[Block,t, Treat] <- InterceptSu[Treat] + b[Treat, 1]
        * muS[Block, t, Treat] + b[Treat, 2] * muF[Block, t-1, Treat] +
        b[Treat, 3] * muS[Block, t-1, Treat] + b[Treat, 4] * muF[Block, t-
2, Treat] + muS[Block, t, Treat] + BlockSu[Block]
      }
    }
  }

  # Combine coefficients: ANNUAL MODEL
  for(Treat in 1:4){

```

```

Omega1[Treat] <- a[Treat, 1] + b[Treat, 1] + a[Treat, 2] +
b[Treat, 2] + a[Treat, 1] * b[Treat, 1]
Omega2[Treat] <- a[Treat, 3] + b[Treat, 3] + a[Treat, 4] +
b[Treat, 4] + a[Treat, 1] * b[Treat, 3] + b[Treat, 1] * a[Treat,
3] - a[Treat, 2] * b[Treat, 2]
}

# test for differences between treatments
deltaInterceptWiTreat12 <- InterceptWi[1] - InterceptWi[2]
deltaInterceptWiTreat13 <- InterceptWi[1] - InterceptWi[3]
deltaInterceptWiTreat23 <- InterceptWi[2] - InterceptWi[3]
deltaInterceptWiTreat14 <- InterceptWi[1] - InterceptWi[4]
deltaInterceptWiTreat24 <- InterceptWi[2] - InterceptWi[4]
deltaInterceptWiTreat34 <- InterceptWi[3] - InterceptWi[4]
deltaInterceptSuTreat12 <- InterceptSu[1] - InterceptSu[2]
deltaInterceptSuTreat13 <- InterceptSu[1] - InterceptSu[3]
deltaInterceptSuTreat23 <- InterceptSu[2] - InterceptSu[3]
deltaInterceptSuTreat14 <- InterceptSu[1] - InterceptSu[4]
deltaInterceptSuTreat24 <- InterceptSu[2] - InterceptSu[4]
deltaInterceptSuTreat34 <- InterceptSu[3] - InterceptSu[4]
deltaOmega1Treat12 <- Omega1[1] - Omega1[2]
deltaOmega1Treat13 <- Omega1[1] - Omega1[3]
deltaOmega1Treat23 <- Omega1[2] - Omega1[3]
deltaOmega1Treat14 <- Omega1[1] - Omega1[4]
deltaOmega1Treat24 <- Omega1[2] - Omega1[4]
deltaOmega1Treat34 <- Omega1[3] - Omega1[4]
deltaOmega2Treat12 <- Omega2[1] - Omega2[2]
deltaOmega2Treat13 <- Omega2[1] - Omega2[3]
deltaOmega2Treat23 <- Omega2[2] - Omega2[3]
deltaOmega2Treat14 <- Omega2[1] - Omega2[4]
deltaOmega2Treat24 <- Omega2[2] - Omega2[4]
deltaOmega2Treat34 <- Omega2[3] - Omega2[4]

# priors for process error
for(Treat in 1:4){
  for(Block in 1:6){
    tauWi[Block, Treat] <- 1 / (sig_pWi[Block, Treat] *
sig_pWi[Block, Treat]) # process error, WINTER
    tauSu[Block, Treat] <- 1 / (sig_pSu[Block, Treat] *
sig_pSu[Block, Treat])
    sig_pWi[Block, Treat] ~ dunif(0, 5) # uniform prior on standard
deviation
    sig_pSu[Block, Treat] ~ dunif(0, 5)
  }
}
# priors for density dependent coefficients
for(i in 1:4){
  for(Treat in 1:4){
    a[Treat, i] ~ dnorm(0, 1)
    b[Treat, i] ~ dnorm(0, 1)
  }
}

```

```
for(Treat in 1:4){  
    InterceptWi[Treat] ~ dnorm(0, 0.0001)  
    InterceptSu[Treat] ~ dnorm(0, 0.0001)  
}  
for(i in 1:6){  
    BlockWi[i] ~ dnorm(0, BlockWi.prec)  
    BlockSu[i] ~ dnorm(0, BlockSu.prec)  
}  
# hyperpriors for Block random effects  
BlockWi.prec <- pow(BlockWi.sd, -2)  
BlockSu.prec <- pow(BlockSu.sd, -2)  
BlockWi.sd ~ dunif(0, 5)  
BlockSu.sd ~ dunif(0, 5)  
}
```

DATA